Diagramme Skalierung Teil 1
Versionen: Excel 2007, 2003, 2002/XP, 2000 und 97

Wenn Sie ein Diagramm mit dem Diagramm-Assistenten erstellen, passt Excel die Skalierungen der X- und Y-Achsen automatisch an. Häufig ist das völlig in Ordnung. Doch es gibt Situationen, in denen die Automatik versagt. Das führt zum Beispiel dazu, dass in Punktdiagrammen die X-Achse bei Null beginnt, obwohl die Werte vielleicht erst bei einem X-Wert von 1000 beginnen. Oder Säulendiagramme sind zu lang und zeigen bei den Spitzenwerten nicht die wesentlichen Nuancen.

Sie müssen also Einfluss auf die Skalierungen der X- und/oder Y-Achsen nehmen. In diesem Tipp erfahren Sie die erforderlichen Details. Und es kommt noch besser: Wir stellen Ihnen eine VBA-Lösung vor, mit der Sie diese Aufgabe vollautomatisch erledigen.

Beginnen Sie mit einem einfachen Beispiel. Dazu geben Sie folgende Beispielwerte in ein Tabellenblatt einer neuen Arbeitsmappe ein:

A1: 1000
A2: 1100
A3: 1200
A4: 1300
A5: 1400
A6: 1500
B1: 5
B2: 10
B3: 13
B4: 14
B5: 15
B6: 15

In Excel 2003, 2002/XP, 2000 und 97 gehen Sie nun folgendermaßen vor: Markieren Sie den Bereich A1:B6 und klicken Sie in der Excel-Symbolleiste auf das Symbol Diagramm-Assistent. Wählen Sie als Diagrammtyp "Punkt (XY)" und als Untertyp die Diagrammart mit interpolierten Linien. Anschließend fahren Sie im Diagramm-Assistenten fort, wobei Sie die Standardwerte immer mit Weiter bestätigen. Auf der letzten Seite aktivieren Sie die Option, mit der das Diagramm als Objekt in das aktuelle Tabellenblatt eingefügt wird. Nach einem Klick auf Fertig stellen bzw. Ende erscheint das Diagramm auf dem Bildschirm.

In Excel 2007 markieren Sie den Bereich A1:B6 und wechseln in der Multifunktionsleiste auf das Register "Einfügen". Anschließend wählen Sie Punkt-Punkte mit interpolierten Linien und Datenpunkten an.

Sie werden feststellen, dass die Linie etwa in der Mitte des Diagramms beginnt. Dabei sind die X-Werte unter 1000 vollkommen uninteressant, da es gar keine zugeordneten Y-Werte gibt. Bei Punktdiagrammen setzt Excel den Koordinatenursprung aber immer auf "0".

Damit im Diagramm nur der wesentliche Bereich angezeigt wird, ändern Sie die Skalierungen der X- und der Y-Achse. Das erreichen Sie in Excel 2003, 2002/XP, 2000 und 97 mit den folgenden Schritten:

1. Klicken Sie einmal auf das Diagrammobjekt im Tabellenblatt, sodass die Symbolleiste Diagramm angezeigt wird.

2. Darin öffnen Sie die Dropdown-Liste am linken Ende der Symbolleiste und wählen den Eintrag "Größenachse (X)" aus. Die X-Achse wird markiert.

3. Klicken Sie in der Diagramm-Symbolleiste auf das Symbol rechts neben dem Dropdown-Feld. Bei der aktuellen Markierung hat es die Bezeichnung Achse formatieren.

4. Im daraufhin angezeigten Dialogfenster wechseln Sie auf die Registerkarte Skalierung.

In Excel 2007 klicken Sie einmal in das Diagramm, um es zu aktivieren und dann mit der rechten Maustaste auf die Beschriftungen der X-Achse. Rufen Sie dann das Kontextmenü Achse formatieren auf. Im nachfolgenden Dialogfeld ist die Rubrik "Achsenoptionen" mit den Einstellungen zur Skalierung eingestellt.

Wie Sie an den Kontrollkästchen erkennen, sind alle Skalierungsparameter auf Automatisch bzw. in Excel 2007 auf Auto gesetzt. Um Minimum und Maximum der X-Achse manuell einzustellen, entfernen Sie die Häkchen in den Feldern Minimum (bzw. Kleinstwert in Excel 97) sowie Maximum (bzw. Höchstwert). In Excel 2007 wechseln Sie die Option Auto jeweils auf Fest. Danach setzen Sie das Minimum auf "1000" und das Maximum auf "1500" - also auf die kleinsten und größten X-Werte aus der zugrunde liegenden Tabelle. Diese Änderungen bestätigen Sie mit Ok bzw. in Excel 2007 mit Schließen.

Zum Skalieren der Y-Achse verfahren Sie prinzipiell genau so. Wiederholen Sie dazu in Excel 2003, 2002/XP, 2000 und 97 die oben beschriebenen Schritte, wobei Sie als Diagrammobjekt die "Größenachse (Y)" im Dropdown-Feld der Symbolleiste auswählen. Bei der Skalierung der Y-Achse müssen Sie natürlich Minimal- und Maximalwerte eingeben, die dafür sinnvoll sind - hier zum Beispiel "5" als Minimum und "15" als Maximum. In Excel 2007 genügt es, die Beschriftungen der Y-Achse mit der rechten Maustaste anzuklicken und wieder das Kontextmenü Achse formatieren aufzurufen.

Im Diagramm sehen Sie anschließend nur noch den Bereich, der relevante Daten enthält.

Nicht in allen Diagrammtypen können Sie die Skalierung so genau beeinflussen. Bei manchen Diagrammarten lässt sich nur die Y-Achse individuell skalieren (etwa bei Säulendiagrammen) und bei manchen Diagrammarten ist eine Skalierung auch gar nicht zweckmäßig; denken Sie zum Beispiel an Kreisdiagramme.

Für ein statisches Diagramm zur einmaligen Verwendung reicht die manuelle Skalierung aus. Aber was ist, wenn Sie die Werte, auf denen das Diagramm basiert, häufig ändern? Oder wenn Sie die Datenreihen laufend erweitern? In manchen Wertereihen ist es außerdem schwierig, auf Anhieb die Minimal- und Maximalwerte zu ermitteln. Und dann kann es noch sein, dass Ihr Diagramm nicht nur eine sondern zwei Y-Achsen enthält - für Datenreihen, deren Werte stark voneinander abweichen.

Leider können Sie im Dialogfenster, in dem Sie die Achsenskalierung vornehmen, keine Formeln, mit denen Sie Minimum und Maximum automatisch berechnen lassen, verwenden. Auch ist es nicht möglich, mit Zellbezügen zu arbeiten, um zum Beispiel auf Zellen mit MIN- oder MAX-Formeln zu verweisen. Aber zum Glück gibt es VBA, die Programmiersprache für Excel. Wie Sie X- und Y-Achsen mit einem VBA-Makro automatisch auf Knopfdruck skalieren, zeigen wir nächste Woche.

Excel skaliert Diagramme nicht immer optimal. Wie Sie die Skalierung selbst beeinflussen, haben Sie im ersten Teil dieses Tipps erfahren. Das manuelle Verfahren kann aber recht zeitaufwändig werden, vor allem wenn Sie die Datenbasis eines Diagramms häufig ändern. Schneller geht es mit einer VBA-Lösung, die wir Ihnen im Anschluss vorstellen möchten.

Nehmen Sie als Beispiel noch einmal die Tabellendaten, die Sie im ersten Teil in den Bereich A1:B6 eingegeben hatten und ergänzen Sie sie durch eine dritte Datenreihe in den Zellen C1:C6:

C1: 60000
C2: 30000
C3: 25000
C4: 24000
C5: 28000
C6: 35000

Markieren Sie dann den gesamten Bereich A1:C6 und erstellen Sie wie im ersten Teil ein neues Punkt (XY)-Diagramm mit interpolierten Linien. Ihnen wird auffallen, dass die Linie der ersten Datenreihe kaum noch Aussagekraft hat, da sich deren Werte sehr stark von der zweiten Datenreihe unterscheiden. In diesem Fall hilft auch eine Skalierung der Y-Achse nicht weiter. Hier bietet es sich an, mit einer zweiten Y-Achse, einer Sekundärachse, zu arbeiten.

Dazu wählen Sie mit Hilfe der Diagramm-Symbolleiste das Objekt "Reihe 2" aus. Anschließend klicken Sie rechts neben dem Objekt-Dropdown-Feld auf das Symbol Datenreihen formatieren. Auf der Registerkarte Achsen aktivieren Sie dann die Option Sekundärachse. Nach einem Klick auf Ok lassen sich nun die Werte beider Datenreihen sinnvoll auswerten.

Es bleibt aber das Problem mit der Achsenskalierung. Wie Sie am Beispiel sehen, müssten Sie inzwischen schon drei Achsen (X-Achse, primäre Y-Achse und sekundäre Y-Achse) bearbeiten. Diesen Aufwand ersparen Sie sich, indem Sie ein VBA-Makro einsetzen:

· Wechseln Sie mit Alt+F11 in den Visual Basic-Editor.

· Rufen Sie Einfügen-Modul auf, um ein neues Codefenster zu öffnen.

· In das Codefenster geben Sie folgenden VBA-Code ein:

Sub AutoAdjustChart()
Dim oChart As Chart
Dim oSeries As Series
Dim aXYSource()
Dim rXAxisSource As Range
Dim rY1AxisSource As Range
Dim rY2AxisSource As Range

Dim i As Long
Dim sFormula As String
Dim dMinVal As Double
Dim dMaxVal As Double

  On Error Resume Next
  Set oChart = ActiveChart
  If oChart Is Nothing Then
    MsgBox "Bitte erst ein Diagramm markieren " & _
        "oder ein Diagrammblatt aktivieren.", vbCritical
    Exit Sub
  End If
  i = 0
  For Each oSeries In oChart.SeriesCollection
    ReDim Preserve aXYSource(2, i)
    sFormula = oSeries.Formula
    aXYSource(0, i) = GetSourceRange(sFormula, xlCategory)
    If oSeries.AxisGroup = xlPrimary Then
      aXYSource(1, i) = GetSourceRange(sFormula, xlValue)
    Else
      aXYSource(2, i) = GetSourceRange(sFormula, xlValue)
    End If
    i = i + 1
  Next oSeries

  Set rXAxisSource = Range(aXYSource(0, 0))
  For i = 0 To UBound(aXYSource, 2)
    Set rXAxisSource = _
        Union(Range(aXYSource(0, i)), rXAxisSource)
    If rXAxisSource Is Nothing Then _
        Set rXAxisSource = Range(aXYSource(0, i + 1))
  Next i
  Set rY1AxisSource = Range(aXYSource(1, 0))
  For i = 0 To UBound(aXYSource, 2)
    Set rY1AxisSource = _
        Union(Range(aXYSource(1, i)), rY1AxisSource)
    If rY1AxisSource Is Nothing Then _
        Set rY1AxisSource = Range(aXYSource(1, i + 1))
  Next i
  If oChart.HasAxis(xlValue, xlSecondary) = True Then
    Set rY2AxisSource = Range(aXYSource(2, 0))
    For i = 0 To UBound(aXYSource, 2)
      Set rY2AxisSource = _
          Union(Range(aXYSource(2, i)), rY2AxisSource)
      If rY2AxisSource Is Nothing Then _
          Set rY2AxisSource = Range(aXYSource(2, i + 1))
    Next i
  End If


  dMinVal = _
  Application.WorksheetFunction.Min(rXAxisSource)

  dMaxVal = _
  Application.WorksheetFunction.Max(rXAxisSource)

  With oChart.Axes(xlCategory)
    .MinimumScale = dMinVal
    .MaximumScale = dMaxVal
  End With

  dMinVal = _
  Application.WorksheetFunction.Min(rY1AxisSource)

  dMaxVal = _
  Application.WorksheetFunction.Max(rY1AxisSource)

  With oChart.Axes(xlValue, xlPrimary)
    .MinimumScale = dMinVal
    .MaximumScale = dMaxVal
  End With
  If oChart.HasAxis(xlValue, xlSecondary) = True Then

    dMinVal = _
    Application.WorksheetFunction.Min(rY2AxisSource)

    dMaxVal = _
    Application.WorksheetFunction.Max(rY2AxisSource)

    With oChart.Axes(xlValue, xlSecondary)
      .MinimumScale = dMinVal
      .MaximumScale = dMaxVal
    End With
  End If

  Set rY2AxisSource = Nothing
  Set rY1AxisSource = Nothing
  Set rXAxisSource = Nothing
  Set oSeries = Nothing
  Set oChart = Nothing
End Sub

'###### HILFSFUNKTION #####
Private Function GetSourceRange( _
    sSeriesFormula As String, _
    iCatVal As Integer) _
    As String
Dim iStartpos As Integer
Dim sTemp As String
  If iCatVal = xlCategory Then
    iStartpos = InStr(1, sSeriesFormula, ",")
    sTemp = Mid(sSeriesFormula, iStartpos + 1, _
        InStr(iStartpos + 1, sSeriesFormula, ",") _
          - iStartpos - 1)
  ElseIf iCatVal = xlValue Then
    iStartpos = InStr(1, sSeriesFormula, ",")
    iStartpos = InStr(iStartpos + 1, sSeriesFormula, ",")
    sTemp = Mid(sSeriesFormula, iStartpos + 1, _
        InStr(iStartpos + 1, sSeriesFormula, ",") _
          - iStartpos - 1)
  Else
    sTemp = ""
  End If
  GetSourceRange = sTemp
End Function
· Verlassen Sie den Visual Basic-Editor per Datei-Schließen und zurück zu Microsoft Excel.

Das Makro ist damit erfasst und einsatzbereit. In der nächsten Woche zeigen wir, wie Sie die neue Funktion nutzen und wie das Makro genau funktioniert.

Excel skaliert Diagramme nicht immer optimal. Wie Sie die Skalierung selbst beeinflussen, haben Sie im ersten Teil dieses Tipps erfahren. Das manuelle Verfahren kann aber recht zeitaufwändig werden, vor allem wenn Sie die Datenbasis eines Diagramms häufig ändern. Schneller geht es mit einer VBA-Lösung, die wir Ihnen dann in der letzten Woche gezeigt haben. In dieser Woche schließen wir unseren Tipp mit Hinweisen zu Einsatz und Funktionsweise des Makros ab.

Um die VBA-Lösung gleich auszuprobieren, markieren Sie das Beispieldiagramm per Mausklick. Danach drücken Sie Alt + F8, womit Sie ein Dialogfenster mit der Liste verfügbarer Makros öffnen. Doppelklicken Sie auf den Makronamen "AutoAdjustChart". Im Nu sind alle Achsen des Diagramms so skaliert, dass nur noch die relevanten Wertebereiche zu sehen sind.

Und wie funktioniert die VBA-Lösung? Die Prozedur analysiert alle Datenreihen des derzeit markierten Diagramms und liest aus der zugrunde liegenden DATENREIHE-Formel die Zelladressen der X- und Y-Werte aus (die Formel erscheint in der Bearbeitungszeile von Excel, sobald Sie eine Datenreihe im Diagramm markieren). Anschließend erstellt die Prozedur separate "Range"-Objekte für die Wertebereiche der X-Achse sowie der primären und der sekundären Y-Achse. Zum Schluss werden die Minimal- und Maximalwerte aus diesen Bereichen jeweils per Tabellenfunktion ermittelt und gleichzeitig als Minimum und Maximum der zugehörigen Diagrammachse eingesetzt.

Um die Zelladressen der X- und Y-Werte zu erhalten, greift die Prozedur auf eine Hilfsfunktion zurück. Das ist nötig, da das Objektmodell von Excel keine Eigenschaft besitzt, die die Datenbasis einer Datenreihe zurückgibt. So muss die bereits angesprochene DATENREIHE-Formel (der VBA-Code arbeitet allerdings mit der internationalen Formel, das heißt mit der SERIES-Funktion) untersucht werden, worum sich die Hilfsfunktion "GetSourceRange" kümmert. Als Parameter übergeben Sie ihr die Formel sowie eine der Konstanten "xlCategory" - zum Auslesen des X-Wertebereichs - oder "xlValue" - für den Y-Wertebereich. Im Beispiel sieht die Formel für die erste Datenreihe etwa so aus:

=SERIES(,Tabelle1!$A$1:$A$6,Tabelle1!$B$1:$B$6,1)

Mit Hilfe der Textfunktionen "InStr" und "Mid" lassen sich dann die Argumente der SERIES-Formel auslesen - hier zum Beispiel "Tabelle1!$A$1:$A$6" für den X-Wertebereich und "Tabelle1!$B$1:$B6" für den Y-Wertebereich.

Da die Wertebereiche mehrerer Diagramm-Datenreihen unterschiedlich groß sein können, muss die Prozedur im nächsten Schritt die Schnittmenge aller Bereiche bilden, die zu einer Achse gehören. Dafür wurden die zuvor ermittelten Wertebereichsadressen in ein zweidimensionales Array ("aXYSource") geschrieben, dessen Positionen nun zu den Gesamt-Wertebereichen für die X-Achse, die primäre Y-Achse und die sekundäre Y-Achse zusammengefasst werden. Als Ergebnisse erhalten Sie jeweils "Range"-Objekte.

In diesen "Range"-Objekten können Sie im letzten Schritt nach den Minimal- und Maximalwerten suchen. Dafür setzen Sie die Tabellenfunktionen "Min" und "Max" ein - zum Beispiel in der Form:

dMinVal = Application.WorksheetFunction.Min(rXAxisSource)

...für den Minimalwert im Bereich der X-Achsenwerte.

Die Ergebnisse weisen Sie anschließend den "MinimumScale"- und "MaximumScale"-Eigenschaften der passenden Diagrammachse zu - hier etwa der primären Y-Achse:

With oChart.Axes(xlValue, xlPrimary)
  .MinimumScale = dMinVal
  .MaximumScale = dMaxVal
End With

Am Bildschirm wird das Diagramm nach diesen Aktionen automatisch aktualisiert, sodass Sie sofort das fertige Resultat sehen.

Beachten Sie aber, dass die Aktionen der VBA-Prozedur nicht rückgängig gemacht werden können. Prüfen Sie also immer an einer Beispieldatei, ob sich die Prozedur in Ihrem konkreten Fall sinnvoll einsetzen lässt.

